The Calendar Program

DEFINE DATA LOCAL

1 #DATE-TO-TEST (A8)

1 REDEFINE #DATE-TO-TEST

  2 #MMP (N2)

  2 #DDP (N2)

  2 #YYP (N4)

1 #MM (A2) INIT <' 1'>

1 #DD (A2)

1 #YY (A4)

1 #WEEK (N2/1:6,1:7)

1 REDEFINE #WEEK

  2 #WEEK-STRING (N2/1:42)

1 #WEEK-LINE (N2)

1 #WEEK-DAY (N2)

1 #DATN (A8)

1 #DAYS-IN-MONTH (N2/1:12)

  CONST <31,28,31,30,31,30,31,31,30,31,30,31>

1 #STARTING-DAY (N1/2000:2010) INIT <7,2,3,4,5,7,1,2,3,5,6>

1 #MONTH-OF-DAYS (N2/1:31)

  INIT <1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,

        16,17,18,19,20,21,22,23,24,25,26,27,28,29,30,31>

1 #START (N4)

1 #END (N4)

1 #END-MINUS-ONE (N4)

1 #MONTH-TEXT (A10/1:12)

  INIT <'JANUARY','FEBRUARY','MARCH','APRIL','MAY','JUNE','JULY',

        'AUGUST','SEPTEMBER','OCTOBER','NOVEMBER','DECEMBER'>

1 #MONTH-YEAR-FOR-MAP (A15)

1 #DATE (D)

1 #DATE-A (A8)

1 #A (A1)

1 #DUMMY (P5)

1 #REM (P5)

END-DEFINE

*

***********************************

FORMAT KD=ON

SET KEY PF2 NAMED 'PICK' PF4 NAMED '<MON' PF6 NAMED '>MON' PF8 NAMED '<YEAR'

        PF10 NAMED '>YEAR'

***********************************

* FIRST, ESTABLISH STARTING YEAR AND MONTH

* THIS WOULD PROBABLY COME FROM RECORD DATA

* OR FROM USER INPUT (SEE HELPROUTINE VARIATION).

*

MOVE 1 TO #MMP

MOVE 5 TO #DDP

MOVE 2005 TO #YYP

*

MOVE *DATN TO #DATN

*

IF #YYP = 0

   MOVE SUBSTRING (#DATN,1,4) TO #YY

   END-IF

*

IF #MMP = 0

   MOVE SUBSTRING (#DATN,5,2) TO #MM

   END-IF

*

IF #DDP = 0

   MOVE SUBSTRING (#DATN,7,2) TO #DD

   END-IF

*

*******************************************

* NOW FIGURE OUT WHAT THAT MONTH LOOKS LIKE

*

PERFORM MONTH-TO-SCREEN

*

*****************************************

DEFINE SUBROUTINE MONTH-TO-SCREEN

COMPRESS #MONTH-TEXT (#MMP) #YYP INTO #MONTH-YEAR-FOR-MAP

*************************************************************

** NOW GET THE STARTING DAY OF WEEK FOR CURRENT YEAR - MONTH

MOVE #MMP TO SUBSTRING (#DATE-A,1,2)

MOVE '01' TO SUBSTRING (#DATE-A,3,2)

MOVE #YYP TO SUBSTRING (#DATE-A,5,4)

MOVE EDITED #DATE-A  TO #DATE (EM=MMDDYYYY)

MOVE EDITED #DATE (EM=O) TO #A

COMPUTE #START = VAL (#A) + 1

IF #START = 8

   MOVE 1 TO #START

   END-IF

COMPUTE #END = #DAYS-IN-MONTH (#MMP)

COMPUTE #END-MINUS-ONE = #END - 1

**

** FIX FOR LEAP YEAR FEBRUARY

*

IF #MMP NE 2

IGNORE

 ELSE

*

   MOVE 29 TO #DDP

   IF #DATE-TO-TEST = MASK (MMDDYYYY)

      ADD 1 TO #END

      ADD 1 TO #END-MINUS-ONE

      END-IF

*

END-IF

*

RESET #WEEK-STRING (*)

MOVE #MONTH-OF-DAYS (1:#END) TO #WEEK-STRING (#START:#START+#END-MINUS-ONE)

ESCAPE ROUTINE

END-SUBROUTINE

**************************************************

* NOW DISPLAY THE STARTING MONTH

*

INPUT (IP=OFF AD=M ZP=OFF)

    1/2 'SUN MON TUE WED THU FRI SAT'

    2/2 #WEEK (1,*)

    3/2 #WEEK (2,*)

    4/2 #WEEK (3,*)

    5/2 #WEEK (4,*)

    6/2 #WEEK (5,*)

    7/2 #WEEK (6,*)

    8/2 #MONTH-YEAR-FOR-MAP (AD=O SG=OFF)

****************************

IF *PF-NAME = '>MON'

   ADD 1 TO #MMP

   IF #MMP GT 12

      COMPUTE #MMP = 12

      REINPUT 'ERROR, MONTH IS DECEMBER CANNOT GO FORWARD'

      END-IF

    PERFORM MONTH-TO-SCREEN

    REINPUT FULL ' '

    END-IF

****************************

IF *PF-NAME = '<MON'

   SUBTRACT 1 FROM #MMP

   IF #MMP LT 1

      COMPUTE #MMP = 1

      REINPUT 'ERROR, MONTH IS JANUARY CANNOT GO BACKWARD'

      END-IF

    PERFORM MONTH-TO-SCREEN

    REINPUT FULL ' '

    END-IF

***********

IF *PF-NAME = '>YEAR'

   ADD 1 TO #YYP

    PERFORM MONTH-TO-SCREEN

    REINPUT FULL ' '

    END-IF

****************************

IF *PF-NAME = '<YEAR'

   SUBTRACT 1 FROM #YYP

    PERFORM MONTH-TO-SCREEN

    REINPUT FULL ' '

    END-IF

****************************

IF *PF-NAME = 'PICK'

   COMPUTE #WEEK-LINE = *CURS-LINE - 1

   COMPUTE #WEEK-DAY = ((*CURS-COL - 2) / 4 ) + 1

   MOVE #WEEK (#WEEK-LINE,#WEEK-DAY) TO #DDP

   WRITE 'YOU HAVE SELECTED' #MMP #DDP #YYP

   END-IF

END

